EXERCICES DE SIMULATION SPICE
MIC4120.- SESSION HIVER 2010

SP1.-CIRCUIT A DIODE

```
.model tut_diode d (is=1e-14 vj=0.6 rs=10)
vs vs gnd 5V
rs vs vd 5k
d1 vd gnd tut_diode
.dc vs 0V 5V 0.01V
.option post=2
.end
```

SP2.- SOURCE PAR IMPULSIONS AVEC DIODE

```
SOURCE PAR IMPULSIONS AVEC DIODE
V1 1 0 PULSE (0 10 5M 0 0 2M 4M)
R1 1 2 100
D1 2 3 DMOD
R2 3 0 1K
C1 3 0 20U
.MODEL DMOD D()
.tran 1U 20M
.PRINT TRAN V(3) V(1)
.END
```

SP3.- RECTIF1

```
VS1 1 0 sin(0V 168V 60)
VS2 0 3 sin(0V 168V 60)
D1 1 2 DMOD
D2 3 2 DMOD
.MODEL DMOD D()
RL 2 0 5
.tran .2m 60m
.PRINT tran V(2) V(1) i(1)
.option post
.END
```

SP4.-FILTRE RC PASSE BAS

```
FILTRE RC PASSE BAS
VS      1       0       AC      1       SIN(0 1   2000)
R1      1       2       1K
C1      2       0       .5U
.AC     DEC     5 10 5000
.TRAN  5US  5M
* RESULTATS
.PRINT  AC      VM(2) VP(2)
```

```
.PLOT   AC       VM(2) VP(2)
.PRINT TRAN   V(1) V(2)
.PLOT   TRAN   V(1) V(2)
.PROBE
.END
```

SP5.-SOURCE AC AVEC DIODE

```
SOURCE AC AVEC  DIODE
*
VI 1 0 SIN(0 10 1KHZ)
C 1 2 10U IC=5v
D 2 3 DMOD
VB 3 0 5V
.MODEL DMOD D()
.TRAN 1U 2M UIC
.PRINT V(2)
.END
```

SP6.- CIRCUIT RLC

```
CIRCUIT RLC
*
VIN 1 0 AC 5 0
L1 2 3 0.125
C1 3 0 1U
R1 1 2 200
.AC DEC 10 1 10k
.PRINT AC IM(VIN) IP(VIN) VM(2) VP(2) VM(3) VP(3)
.PLOT AC VDB(3) VP(3)
.END
```

SP7.-ZENER1

```
VS1 0 1 12
R1 1 2 5
DZ1 2 0 DMOD
.MODEL DMOD D(BV=10 IBV=100u)
RL 2 0 200
.TRAN 1U 2M
.PRINT TRAN V(2)
.END
```

SP8.- ZENER SPIKES

```
*
VS 1 0 SIN(0V 10V 1KHZ)
VP 2 1 PULSE (0V 10V 0.25M 0.01M 0.01M 1U 1M)
R 2 3 1
```

```
d1 4 3 DMOD
d2 4 0 DMOD
rl 3 0 50
.MODEL DMOD D(BV=9.3V IBV=100u)
.TRAN 1U 2M
.PRINT V(3)
.END
```

SP9.- ZENERLIMITTER

```
*
Vi 1 0 DC 0V
R 1 2 1k
D1 2 0 D1N4148
D2 0 2 D1N4148
.model D1N4148 D (Is=0.1pA Rs=16 CJO=2p Tt=12n Bv=100 Ibv=0.1p)
.DC Vi -5V 5V 100mV
.PLOT DC V(2)
.PRINT V(2)
.END
```

SP10.- FILTRE

```
*
VS 1 0 SIN(0 161 60)
D 1 2 DMOD
.MODEL DMOD D()
L1 2 3 0.05
C1 3 0 2000U
RL 3 0 120
.TRAN .5M 500M
.print TRAN VM(3) VM(1)
.END
```

SP11.- CIRCUIT DOUBLEUR DE TENSION

```
** Circuit Description **
Vi 1 0 sin ( 0 10V 1kHz )
C1 1 2 1u
C2 3 0 1u
D1 2 0 D1N4148
D2 3 2 D1N4148
* diode model statement
.model D1N4148 D (Is=0.1pA Rs=16 CJO=2p Tt=12n Bv=100 Ibv=0.1p)

** Analysis Requests **
.TRAN 100u 10m 0m 100u
** Output Requests **
.PLOT TRAN V(1) V(2) V(3)
```

```
.probe
.end
```

SP12.- SOUS-CIRCUIT

```
EXEMPLE DE SOUS-CIRCUIT
*
X1 1 2 5 INVERTER
X2 2 3 5 INVERTER
X3 3 1 5 INVERTER
VDD 5 0 5

.SUBCKT INVERTER 1 2 3
M1 2 1 0 0 ENH L=10U W=40U
M2 3 2 2 0 DEP L=10U W=5U
.MODEL DEP NMOS LEVEL=1 VTO=-3 LAMBDA=.001 KP=.4E-4
.MODEL ENH NMOS LEVEL=1 VTO=1.8 CGSO=20N LAMBDA=.001 KP=.4E-4
.ENDS INVERTER

.IC V(1)=5 V(2)=0
.TRAN .01U .5U
.PLOT TRAN V(1) V(2) V(3) (0,5)
.WIDTH OUT=80
.END
```

SP13.- TRANSISTOR

```
transist2
*
vs 1 0 AC=1 sin(0v 10mV 1kHz)
CC1 1 2 100U
VEE 0 4 4
VCC 5 0 15
RE 2 4 3.3K
RC 3 5 8.1K
Q1 3 0 2 QMOD
CC2 3 6 100u
RL 6 0 15k
.MODEL QMOD NPN (Is=10fA Ikf=150mA Bf=150 Br=3 Rb=1 Va=30V)
.OP
.dc vcc 15v 15v 1v
.AC DEC 10 1 10K
.PRINT DC V(3,2)
.print ac V(3)
.TRAN 1U 1M
.measure TRAN pow FIND PAR('V(5)*I(vcc)') AT 0.2M
.print tran V(3.2)
.END
```

SP14.- EXEMPLE DE BJT

```
.OPTION       $ The options are listed below:
+ ingold=1    $ Use fixed-point and exponential form on output
+ numdgt=4    $ print 4 digits

*.INC /nfs/guille/analog1/m/moon/ece323/npn322.inc


Vcc    vcc    0     =12.0
q1     c      b     e       npn322
Rb1    vcc    b     2.1k
Rb2    b      0     300
Rc     vcc    c     100
Re1    e      e1    9.5
Re2    e1     0     6.5
Cout   e1     0     63u
Cin    vi     b     1u
Vin    vi     0     AC=1 SIN 0 amp 100K 0 0 180
.model npn322 npn ()

.OP

.AC DEC 10 1 1G
.NET V(c) Vin
.PRINT AC Vdb(c)

.MEASURE AC A0 FIND VR(c) AT=100K                $ Measure the gain at 10kHz.
.MEASURE AC AdB FIND VdB(c) AT=100K              $ Measure the gain at 10kHz.
.MEASURE AC mzi FIND PAR('v(vi)/i(vin)') AT=100K $ Measure the input impe
.MEASURE AC mzo FIND ZOUT(M) AT=100K             $ Measure the out. imp. @10kHz.
.MEASURE AC fl WHEN VM(c)='.707*ABS(A0)'         $ Find the lower 3dB frequency.
.MEASURE AC fu WHEN VM(c)='.707*ABS(A0)' TD=1e5      $ Find the upper 3dB frequency.
.MEASURE AC POW FIND PAR('v(vi)*i(vin)') AT 100K

.TRAN .1u 20u SWEEP amp POINTS 2 110m 500m
.PRINT TRAN V(c)

.END
```

SP15.-CMOS AMPLIFIER

```
* dc supplies
Vdd 1 0 DC +10V
Iref 2 0 DC 100uA
```

```
* input signal
Vi 4 0 DC 0V
* amplifier circuit
M1 3 4 0 0 nmos L=10u W=100u
M2 3 2 1 1 pmos L=10u W=100u
M3 2 2 1 1 pmos L=10u W=100u
* mosfet model statements (by default, level 1)
.model nmos nmos (kp=20u Vto=+1V lambda=0.01)
.model pmos pmos (kp=10u Vto=-1V lambda=0.01)
** Analysis Requests **
* calculate DC transfer characteristics
.DC Vi 0V +10V 10mV
** Output Requests **
.PLOT DC V(3)
.probe
.end
```

SP16.- ** The CMOS Inverter **

```
* dc supplies
Vdd 1 0 DC +5V
* input digital signal
Vi  3 0 DC +5V
* MOSFET inverter circuit
M1 2 3 0 0 MN L=3um W=3um
M2 2 3 1 1 MP L=3um W=9um
* BNR 3um transistor model statements (level 3)
.MODEL MN nmos level=3 vto=.7 kp=4.e-05 gamma=1.1 phi=.6
+       lambda=.01 rd=40 rs=40  pb=.7 cgso=3.e-10 cgdo=3.e-10
+       cgbo=5.e-10 rsh=25 cj=.00044  mj=.5 cjsw=4.e-10 mjsw=.3
+       js=1.e-05 tox=5.e-08 nsub=1.7e+16  nss=0 nfs=0 tpg=1 xj=6.e-07
+       ld=3.5e-07 uo=775  vmax=100000  theta=.11  eta=.05 kappa=1
.MODEL MP pmos level=3 vto=-.8 kp=1.2e-05 gamma=.6 phi=.6
+       lambda=.03 rd=100 rs=100  pb=.6 cgso=2.5e-10 cgdo=2.5e-10
+       cgbo=5.e-10 rsh=80 cj=.00015  mj=.6 cjsw=4.e-10 mjsw=.6
+       js=1.e-05 tox=5.e-08 nsub=5.e+15  nss=0 nfs=0 tpg=1 xj=5.e-07
+       ld=2.5e-07 uo=250  vmax=70000  theta=.13  eta=.3 kappa=1
** Analysis Requests **
.DC Vi 0 5 50mV
** Output Requests **
.PLOT DC V(2) Id(M1)
.probe
.end
```

SP17.- ** A Simple Operational Amplifier **

```
* store the simulation results requested in a graph data file
.options post
```

```
** Circuit Description **
* power supplies
Vcc 4 0 DC +15V
Vee 5 0 DC -15V
* differential-mode signal level
Vd 101 0 DC 0V
* for small-signal parameters, bias with -Vos across input terminals
* Vd 101 0 DC -178.5uV AC 1
Rd 101 0 1
EV+ 1 100 101 0 +0.5
EV- 2 100 101 0 -0.5
* common-mode signal level
Vcm 100 0 DC 0V
* 1st stage
R1 4 7 20k
R2 4 8 20k
Q1 7 1 6 npn_transistor
Q2 8 2 6 npn_transistor
Q3 6 9 5 npn_transistor
* 2nd stage
R3 4 11 3k
Q4 4 7 10 npn_transistor
Q5 11 8 10 npn_transistor
Q6 10 9 5 npn_transistor 4
* 3rd or output stage
R4 4 12 2.3k
Q7 13 11 12 pnp_transistor
R5 13 5 15.7k
Q8 4 13 3 npn_transistor
R6 3 5 3k
* biasing stage
Rb 0 9 28.6k
Q9 9 9 5 npn_transistor
* transistor model statements
.model npn_transistor npn ( Is=1.8fA Bf=100 VAf=100V )
.model pnp_transistor pnp ( Is=1.8fA Bf=100 VAf=100V )
** Analysis Requests **
* compute DC operating point using the following initial guesses
.OP
.NODESET V(3)=0V V(6)=-0.7V V(7)=+10V V(8)=+10V V(9)=-14.3V V(10)=+9.3V
+       V(11)=+12V V(12)=+12.7V V(13)=+0.7V
* compute large-signal differential-input transfer characteristics of amp
.DC Vd -15V +15V 100mV
* expanded view of high-gain region
* .DC Vd -5mV +5mV 10uV
** Output Requests **
.PLOT DC V(3)
```

```
* compute the small-signal parameters of the amplifier
* .tf v(3) vd
* .ac lin 1 1Hz 1Hz
* .print ac Im(ev+) Im(ev-) vm(1,2)
* for input common-mode range (with Vd=-178.5uV)
* .DC Vcm -15V +15V 0.1V
.end
```

SP18.- ** A Common-Emitter Amplifier **

```
* power supplies
Vcc 1 0 DC +12V
* input signal source
Vs 7 0 AC 1V
Rs 7 6 4k
* CE stage
Cc1 6 3 1uF
R1 1 3 8k
R2 3 0 4k
Q1 2 3 4 Q2N3904
Re 4 0 3.3k
Rc 1 2 6k
Ce 4 0 10uF
Cc2 2 5 1uF
* output load
Rl 5 0 4k
*
* transistor model statement for 2N3904
.model Q2N3904  NPN (Is=6.734f Xti=3 Eg=1.11 Vaf=74.03 Bf=416.4 Ne=1.259
+       Ise=6.734f Ikf=66.78m Xtb=1.5 Br=.7371 Nc=2 Isc=0 Ikr=0 Rc=1
+       Cjc=3.638p Mjc=.3085 Vjc=.75 Fc=.5 Cje=4.493p Mje=.2593 Vje=.75
+       Tr=239.5n Tf=301.2p Itf=.4 Vtf=4 Xtf=2 Rb=10)
** Analysis Requests **
.OP
.AC DEC 10 1Hz 100MegHz
** Output Requests **
.PLOT AC VdB(5)
.probe
.end
```

SP19.- ** A Cascode Amplifier **

```
* power supplies
Vcc 1 0 DC +15V
* input signal source
Vs 9 0 AC 1V
Rs 9 8 4k
```

```
* CE stage (input stage)
Cc1 6 8 1uF
R1 1 3 18k
R2 3 6 4k
R3 6 0 8k
Q1 4 6 7 Q2N3904
Re 7 0 3.3k
Ce 7 0 10uF
* CB stage (upper stage)
Q2 2 3 4 Q2N3904
Rc 1 2 6k
Cb 3 0 10uF
Cc2 2 5 1uF
* output load
Rl 5 0 4k
*
* transistor model statement for 2N3904
.model Q2N3904  NPN (Is=6.734f Xti=3 Eg=1.11 Vaf=74.03 Bf=416.4 Ne=1.259
+       Ise=6.734f Ikf=66.78m Xtb=1.5 Br=.7371 Nc=2 Isc=0 Ikr=0 Rc=1
+       Cjc=3.638p Mjc=.3085 Vjc=.75 Fc=.5 Cje=4.493p Mje=.2593 Vje=.75
+       Tr=239.5n Tf=301.2p Itf=.4 Vtf=4 Xtf=2 Rb=10)
** Analysis Requests **
.OP
.AC DEC 10 1Hz 100MegHz
** Output Requests **
.PLOT AC VdB(5)
.probe
.end
```

SP20- ** Differential Amplifier With Emitter Resistors **

```
* set acout=0 so HSPICE calculates vdb(x,y) as "magnitude of difference",
* rather than "difference of magnitude"
.options post acout=0

** Circuit Description **
* power supplies
Vcc 1 0 DC +10V
* input signal source
Vs 6 0 AC 1V
Rs 6 5 5k
* differential pair
Rc1 1 2 10k
Rc2 1 3 10k
Q1 3 5 7 Q2N3904
```

```
Q2 2 77 8 Q2N3904
Rs2 77 0 5k
* emitter resistors
* zero-resistance
Re1 7 4 0.00001
Re2 8 4 0.00001
* current biasing
Ibias 4 0 1mA
*
* transistor model statement for 2N3904
.model Q2N3904  NPN (Is=6.734f Xti=3 Eg=1.11 Vaf=74.03 Bf=416.4 Ne=1.259
+       Ise=6.734f Ikf=66.78m Xtb=1.5 Br=.7371 Nc=2 Isc=0 Ikr=0 Rc=1
+       Cjc=3.638p Mjc=.3085 Vjc=.75 Fc=.5 Cje=4.493p Mje=.2593 Vje=.75
+       Tr=239.5n Tf=301.2p Itf=.4 Vtf=4 Xtf=2 Rb=10)
** Analysis Requests **
.OP
.ac DEC 10 1Hz 100MegHz
** Output Requests **
.print vdb(3,2)

* re-run simulation for resistor values 100, 200 and 300
.alter
Re1 7 4 100
Re2 8 4 100

.alter
Re1 7 4 200
Re2 8 4 200

.alter
Re1 7 4 300
Re2 8 4 300
.end
```

SP21.- CE AMPLIFIER

```
* VS 1 0 AC 5 0
vs 1 0 sin(0v 4v 10k)
RS 1 2 100K
CC1 2 3 100U
Q1 4 3 0 QNPN
RF 3 4 180K
RC 4 5 2K
VCC 5 0 12V
CC2 4 6 100U
RL 6 0 2K
.MODEL QNPN NPN()
.DC VCC 12V 12V 1V
```

```
*.AC DEC 10 1 10K
*.PRINT AC IM(VS) IP(VS) V(3) V(4)
.TRAN 1U .1M
.PRINT DC V(3) V(4)
*.PLOT AC VDB(3) VP(3)
.print tran V(3) V(4) IM(VS)
*.probe
.END
```

SP22.- ** Common-Emitter Amplifier Stage **

```
** HSPICE Circuit Description **
* power supplies
Vcc 1 0 DC +10V
Vee 8 0 DC -10V
* input signal
* for output resistance, change Vs from 10mV to 0V
Vs 6 0 AC 10mV
Rs 5 6 10k
* amplifier
C1 4 5 1GF
Rb 4 0 100k
XQ1 2 4 3 t2n2222a
Rc 1 2 10k
Re 3 8 10k
C2 2 7 1GF
C3 3 10 1GF
* emitter degeneration resistor (vary as needed)
Rdegen 10 0 100
* load + ammeter
Rl 7 9 10k
* for output resistance, change Vout from 0 to 10mV
* Vout 9 0 AC 10mV
Vout 9 0 0
* HSpice transistor model statement for the 2N2222A
* betaf=100 in this example: also vary to 153 and 200
 .macro t2n2222a 1 2 3 betaf=100 tauf=5.1e-10
 *02-870918
 q2n2222a 1 2 3 t2n2222a
  .model t2n2222a npn
  + iss    =   0.      xtf    =   1.      ns     =   1.00000
  + cjs    =   0.      vjs    =   0.50000  ptf    =   0.
  + mjs    =   0.      eg     =   1.10000  af     =   1.
  + itf    =   0.50000  vtf    =   1.00000  bf     =  betaf
  + br     =   40.00000  is     =  1.6339e-14 vaf    =  103.40529
  + var    =   17.77498  ikf    =   1.00000  ise    =  4.6956e-15
```

```
 + ne     =    1.31919  ikr    =    1.00000  isc    =  3.6856e-13
 + nc     =    1.10024  irb    =  4.3646e-05 nf     =    1.00531
 + nr     =    1.00688  rbm    =  1.0000e-02 rb     =   71.82988
 + rc     =    0.42753  re     =  3.0503e-03 mje    =    0.32339
 + mjc    =    0.34700  vje    =    0.67373  vjc    =    0.47372
 + tf     =    tauf     tr     =  380.00e-9  cje    =  2.6734e-11
 + cjc    =  1.4040e-11 fc     =    0.95000  xcjc   =    0.94518
 + subs   =    1
 .eom
** Analysis Requests **
* calculate DC bias point information
.OP
.AC LIN 10 .5kHz 10kHz
** Output Requests **
*  voltage gain Av=Vo/Vs
.PRINT AC Vm(6) Vp(6) Vm(7) Vp(7)
*  current gain Ai=Io/Ii
.PRINT AC Im(Vs) Ip(Vs) Im(Vout) Ip(Vout)
* input resistance Ri=Vi/Ii
.PRINT AC Vm(4) Vp(4) Im(Vs) Ip(Vs)
* for output resistance, print these values too
* .PRINT AC Vm(7) Vp(7) Im(Vout) Ip(Vout)
.end
```